


ATARI
MUSEUM.NL
An Waalgreen

Ik en mijn Atari

Wolters-Noordhoff





Ik en mijn Atari





Ik en mijn Atari

Basic voor kinderen

An Walgreen

Wolters-Noordhoff Groningen



Omslagtekening Sylvia Weve

0 1 2 3 4 5 / 88 87 86 85 84

Copyright © 1984 Wolters-Noordhoff bv Groningen,
The Netherlands.

Alle rechten voorbehouden. Niets uit deze uitgave mag worden verveelvoudigd, opgeslagen in een geautomatiseerd gegevensbestand, of openbaar gemaakt, in enige vorm of op enige wijze, hetzij elektronisch, mechanisch, door fotokopieën, opnamen, of op enige andere manier, zonder voorafgaande schriftelijke toestemming van de uitgever.

All rights reserved. No part of this publication may be reproduced, stored in a retrieval system, or transmitted, in any form or by any means, electronic, mechanical, photocopying, recording, or otherwise, without the prior written permission of the publisher.

ISBN 9001 41940 2

Aan de kinderen 6

- 1 Kennismaking met de computer 7
NEW - PRINT - REM - RUN
 - 2 Verbeteren en veranderen 13
 - 3 Een paar trucjes 16
 - 4 Een programma veranderen 21
LIST
 - 5 Hoe je iets in het geheugen stopt 24
DIM - LET - INPUT
 - 6 Springen en stoppen 29
GOTO
 - 7 De IF...THEN instructie 33
IF...THEN
 - 8 Toeters en bellen 37
FOR...NEXT - SOUND
 - 9 Werken met getallen 41
 - 10 Getallen in de IF...THEN instructie 46
 - 11 Toevallige getallen en de functie INT 51
RND - INT
 - 12 Het opslaan van gegevens 55
DATA - READ - RESTORE
 - 13 Kleuren 60
SETCOLOR
 - 14 Programma's bewaren op cassettebandjes 64
CSAVE - CLOAD
- Voorbeeld-programma's 68
Foutmeldingen 72
Spelletjes 73

Aan de kinderen

Dit boekje is speciaal voor jullie geschreven. In dit boek leer je hoe je zelf programma's kunt maken op je ATARI home-computer. In het begin lijkt het je misschien wel wat moeilijk, maar het valt heus wel mee.

De 'taal' waarmee we met de computer kunnen 'praten' heet BASIC.

Deze taal is door mensen uit Amerika bedacht. Daarom zijn de opdrachten die de computer 'begrijpt' ook in het Engels. Deze woorden worden je in dit boekje uitgelegd. We hebben geprobeerd om de programma's in dit boekje zo leuk mogelijk te maken. Er staan spelletjes in die je alleen of met een vriend of vriendin kunt spelen.

Probeer alle opdrachten die bij elk hoofdstuk staan te maken. Als het niet meteen lukt, kijk dan eerst het hoofdstuk nog even goed na. Lukt het dan nóg niet, vraag dan of je vader, je moeder of je onderwijzer(es) je wil helpen. Of kijk achterin het boek. Bij moeilijke opdrachten staat daar een voorbeeld.

Denk erom, de computer kan zelf helemaal niets. ALLES wat hij doet heb JIJ er eerst ingestopt.

Ik hoop dat je het gauw zult leren en wens je veel plezier bij het 'computeren'.

1



Kennismaking met de computer

-
- Je naam typen
 - De DELETE BACK SPACE toets
 - De CAPS toets
 - Het scherm schoonmaken
 - Opdrachten voor de computer
 - Het geheugen leeg maken
 - Hoe maak je een regel?
 - Het getal 0 en de letter 'O'
 - Een programma typen, uitvoeren en begrijpen
 - Hoe nummer je regels?
-

Toen je voor het eerst de ATARI zag heb je misschien gedacht: 'Is dat nou alles? Het lijkt net een gewone schrijfmachine.'

Maar het is veel meer dan dat.

Stel je de computer voor als een grote kast. Op elke plank in die kast staan dozen. Die dozen zijn het geheugen van de computer.

Sommige dozen zijn door de makers van ATARI al gevuld. De andere dozen zijn leeg. Daar kun je zelf iets in stoppen.

Dat doe je in computertaal. Die taal gaan we je in dit boek leren. Dat is niet zo heel moeilijk, want de computer 'begrijpt' maar ongeveer 80 woorden.

Laten we nu dan maar beginnen.

Zet de computer en de tv aan. Op het scherm zie je nu het woord: READY.

Daaronder zie je een wit vierkantje: 'de loper'. Dit vierkantje 'loopt' over het scherm en laat je zien waar de volgende letter komt te staan.

Ready betekent: klaar.

READY met daaronder het vierkantje betekent: je kunt beginnen.

► Je naam typen

Typ Typ je naam maar eens.

Alles wat je typt zie je meteen op het scherm.

Maak nu de regel vol en typ gewoon verder. Je ziet dat de looper vanzelf op een nieuwe regel begint.

Als je een verkeerde letter hebt getypt kun je deze verbeteren: druk op de toets DELETE BACK SPACE.

Deze toets zit rechtsboven op je toetsenbord. De letter die je het laatst hebt getypt, wordt dan uitgedumd. Je kunt nu de goede letter typen.

Rechts op je toetsenbord zit een toets waarop CAPS staat. CAPS is de afkorting van het Engelse woord capitals.

Capitals betekent: hoofdletters.

Als je één keer op deze toets drukt, schrijft de computer kleine letters. Druk je nog eens, dan schrijft de computer weer hoofdletters. In dit boek gebruiken we alleen hoofdletters.

Let op! Als je een instructie in kleine letters typt zegt de computer: ERROR-
ERR●R betekent: fout.

► Het scherm schoonmaken

Om het scherm weer schoon te maken houd je de CONTROL toets ingedrukt. Druk nu op de CLEAR toets. Het scherm is nu weer schoon.

CLEAR betekent: schoon.

CONTROL + CLEAR maken samen het scherm schoon.



► **Opdrachten voor de computer**

Als je de computer iets wilt laten doen, moet je hem opdrachten geven. Deze opdrachten noemen we *instructies*.

Je moet leren welke instructies de computer begrijpt.

► **Het geheugen leeg maken**

Om het geheugen van de computer leeg te maken typ je `NEW` en druk je op de `RETURN` toets. `NEW` maakt het geheugen van de computer leeg. Je kunt er nu een programma instoppen. `NEW` maakt het scherm niet schoon. Alles wat er staat blijft dus staan.

► **Wat is een programma?**

Een programma is een lijst met instructies en gegevens voor de computer. Deze instructies en gegevens worden in regels geschreven. Iedere regel heeft een eigen regelnummer.

► **Hoe maak je een regel?**

Voorbeeld `10 PRINT "HALLO"`

Om een regel te maken doe je de volgende dingen:

- Typ een regelnummer en daarna een spatie.
- Typ `PRINT` en daarna een spatie.
- Typ `"HALLO"`. Denk om de aanhalingstekens!
- Druk op de `RETURN` toets.

`RETURN` betekent: ga terug. `RETURN` doet twee dingen:

- De looper springt weer vooraan op een nieuwe regel.
- De regel die je net getypt hebt, wordt in het geheugen van de computer gestopt.

Nu zit regel 10 in het geheugen van de computer. Daar blijft hij zitten tot je de instructie NEW geeft, of tot je de computer uitzet.

Je hebt net deze regel in het geheugen van de computer gestopt: 10 PRINT "HALLO"

Je hebt het getypt en op de RETURN toets gedrukt.

Typ nu RUN en druk weer op de RETURN toets. De computer heeft HALLO op het scherm geschreven. Daaronder schrijft hij READY. De computer is klaar voor de volgende opdracht.

Wat is er gebeurd?

De instructie RUN zegt tegen de computer: Kijk in je geheugen of er een programma in zit. Ja? Doe dan wat er in dat programma staat.

- De computer moet de PRINT instructie uitvoeren. De PRINT instructie geeft de computer opdracht om alles wat tussen aanhalingstekens staat op het scherm te schrijven.
- De computer schrijft: HALLO
RUN betekent hier: Voer het programma uit.
PRINT betekent hier: Schrijf op het scherm.

Let op! Typ altijd precies wat er in de programmaregels staat. Doe je dit niet, dan zegt de computer ERROR- (fout). Na elke programmaregel moet je op de RETURN toets drukken.

► Het getal 0 en de letter 'O'

De computer maakt een nul altijd zo: 0

En de letter O altijd zo: O

Let erop dat je dit altijd goed doet.

Anders zegt de computer: ERROR-

Bijvoorbeeld:

Goed: 10 PRINT "HALLO"

Fout: 1O PRINT "HALLO"



► **Een programma typen, uitvoeren en begrijpen**

Maak het geheugen leeg met de instructie `NEW`. Denk aan de `RETURN` toets.

Typ Typ nu dit programma:

```
1 REM EERSTE PROGRAMMA
2 PRINT " HALLO "
3 PRINT " ALLEMAAL "
RUN
```

Wat gebeurt er met de regels?

- Regel 1 – De computer slaat deze regel over. De computer doet niets met een `REM`.
`REM` is de afkorting van het Engelse woord *remark*.
Remark betekent: opmerking of notitie.
Met een `REM` maak je kleine notities of opmerkingen voor jezelf in een programma. Hier is een `REM` gebruikt om het programma een naam te geven.
- Regel 2 – De computer schrijft (`PRINT`) `HALLO` op het scherm.
- Regel 3 – De computer schrijft (`PRINT`) `ALLEMAAL` op het scherm.
- `RUN` – De computer schrijft het programma op het scherm, zonder regelnummers.

De computer voert de instructies regel voor regel uit. Hij begint met het laagste regelnummer (1) en gaat in volgorde verder (2, 3).

► **Hoe nummer je regels?**

In het vorige programma hadden we de regels netjes in volgorde genummerd. Als je zelf een programma maakt sla je meestal een paar nummers over. Kijk maar:

Voorbeeld 1 0 REM EERSTE PROGRAMMA
2 0 PRINT " HALLO "
3 0 PRINT " ALLEMAAL "

Het is hetzelfde programma, maar we hebben andere regelnummers gebruikt. De regels zijn nog steeds in volgorde, maar we hebben nummers overgeslagen. Dat is handig, want dan kun je er later nieuwe regels tussen zetten. Bijvoorbeeld als je een programma wilt veranderen.

► **Opdracht**

Maak eerst het geheugen leeg en het scherm schoon.

- 1 Maak zelf een klein programma.
- Gebruik de PRINT instructie om:
je naam, je adres, je woonplaats, je hobby's en in welke klas je zit op het scherm te schrijven. (De komma in hobby's zit boven de 7.)



Verbeteren en veranderen

-
- Tekens herhalen
 - De PIJL toetsen
 - De DELETE toets
 - De INSERT toets
-

► **Tekens herhalen**

Typ de letter A en houd de toets ingedrukt. Je krijgt nu een hele rij met A's.

Dit gaat met elke toets. Probeer maar.

► **De PIJL toetsen**

Rechts op het toetsenbord zitten vier toetsen waarop een pijltje staat. Eén wijst naar boven, één naar beneden, één naar rechts en één wijst naar links.

Houd de CONTROL toets ingedrukt en druk een keer op de toets met het pijltje dat naar beneden wijst. De looper springt nu één plaats naar beneden. Houd de PIJL toets ingedrukt. De looper rent nu naar beneden. Probeer dit ook met de andere PIJL toetsen.

Door CONTROL en een PIJL toets in te drukken kun je de looper op elke plaats op het scherm brengen. Dit is gemakkelijk als je iets wilt veranderen of verbeteren.

Breng de looper weer terug op zijn oude plaats. Je kunt ook op de RESET toets drukken.



Typ 1 0 R E M V O U T E N V E R B E T E L E N

Breng met behulp van de PIJL toetsen de looper op de letter V en typ er een F voor in de plaats. Ga nu met de looper naar de letter L en maak er een R van.

De fouten zijn er uit en er staat nu:

1 0 R E M F O U T E N V E R B E T E R E N

Druk op RETURN om de regel in het geheugen van de computer te stoppen.

► **De DELETE toets**

DELETE betekent: weghalen.

Met deze toets kun je één of meer verkeerde letters uit een woord halen.

Typ 1 5 R E M A R M B R A N D

en druk op RETURN. Zoals je ziet staat er een R te veel in het woord. Ga met de PIJL toets naar de letter R. Houd de CONTROL toets ingedrukt en druk op de DELETE toets. De letter R is weg en de rest van het woord sluit netjes aan. Druk je nóg een keer op de DELETE toets, dan verdwijnt ook de letter A en staat er:

1 5 R E M A R M B N D

► **De INSERT toets**

INSERT betekent: ertussen voegen.

Met deze toets kun je één of meer letters die je hebt vergeten er weer tussen zetten.

In je programma staat:

1 5 R E M A R M B N D



Breng de looper op de letter na de plaats waar je de nieuwe letter wilt invoegen. Hier is dat de letter N. Houd de CONTROL toets ingedrukt en druk op de INSERT toets. Er komt nu een opening links van de letter N en de looper springt op de open plaats. Typ nu de A en er staat:

```
1 5 R E M A R M B A N D
```

Als je de CONTROL en de INSERT toetsen ingedrukt houdt komt er een hele rij openingen. De rij sluit zich weer als je de CONTROL toets en de DELETE toets ingedrukt houdt.

► Opdrachten

Maak het geheugen leeg en het scherm schoon.

- 2 Typ de volgende regel:

```
1 0 P R I N T " I K K O M P U T E L O P J O U W A T E L I "
```

- Maak hiervan met behulp van de DELETE, de INSERT en de PIJL toetsen de volgende zin:

```
1 0 P R I N T " I K C O M P U T E R O P M I J N A T A R I "
```

3. Probeer of je met de letter X in een paar PRINT regels een tekening op het scherm kunt zetten. Bijvoorbeeld een gezicht, een auto of een vogel.

3

Een paar trucjes



- Een regel overslaan
- Zwart-op-wit typen
- De zoemer
- De ESC toets met "clear"
- De ESC toets met "zoem"
- Vaste tekst

Let op! Voor je een nieuw programma gaat typen maak je altijd eerst:

- het geheugen leeg en
- het scherm schoon.


► Een regel overslaan

Typ

```
10 REM REGELS OVERSLAAN
20 PRINT "DIT IS EEN REGEL "
25 PRINT
30 PRINT " IK HEB EEN REGEL OVERGESLAGE
N "
RUN
```

Zie je wel, op de plaats van regel 25 staat nu niets. PRINT zonder iets erachter maakt een lege regel. Dit kan soms mooier staan in je programma's.

► Zwart-op-wit typen

Er zit een speciale toets op je ATARI. Hij zit rechtsonder op je toetsenbord. Hij ziet er zo uit: 

We noemen hem de zwart-op-wit toets. Druk deze toets één keer in en typ dan je naam.

Zie je wat er is gebeurd? Je naam staat nu in zwart-op-wit op het scherm.

Druk nu nog eens op de zwart-op-wit toets. Typ weer je naam. Nu staat alles weer in gewone letters (wit-op-zwart). Je kunt deze toets gebruiken om er je programma leuker mee te maken.

Let op! Typ *nóóit* een instructie in zwart-op-wit. De computer zegt dan `ERROR-` (fout).

- Soms weet je niet wat je fout gedaan hebt als de computer `ERROR-` zegt. Kijk dan nog eens goed naar je programma. Misschien heb je een woord overgeslagen, of de aanhalings-tekens vergeten. Kun je de fout niet vinden, maak dan het geheugen leeg en begin overnieuw.

► De zoemer

De `ATARI` heeft een zoemer. Houd de `CONTROL` toets ingedrukt en druk op de toets waar de `2` op staat. Je hoort nu een zoemer.

`CONTROL + 2` laten samen de zoemer gaan. Zet, als je de zoemer niet goed hebt gehoord, het geluid van de tv wat harder.

► De `ESC` toets met "clear"

Maak eerst het geheugen leeg en het scherm schoon.

Zoek de `ESC` toets. Hij zit in de linkerbovenhoek. `ESC` is de afkorting van het Engelse woord *escape*.

Escape betekent: ontsnappen.

We blijven de afkorting `ESC` gebruiken omdat dit zo op het toetsenbord staat. Met de `ESC` toets kunnen we de computer opdrachten geven.

Telkens wanneer je in een `PRINT` regel het woord "clear" ziet staan (in kleine letters en tussen aanhalingstekens), doe je dit:

Druk één keer op de `ESC` toets. Houd de `CONTROL` toets ingedrukt en druk op de `CLEAR` toets.

Gesnapt? Vergeet niet de aanhalingstekens te typen.

```
Typ 1 0 REM DE ESC TOETS
    1 5 PRINT " clear "
    2 0 PRINT " IK STA OP EEN SCHOON SCHERM "
```

In regel 15 zie je nu niet het woord "clear" op het scherm. In plaats daarvan zie je een krom pijltje.

► **De `ESC` toets met "zoem"**

Telkens wanneer je in een `PRINT` regel het woord "zoem" ziet staan, in kleine letters en tussen aanhalingstekens, doe je dit: Druk één keer op de `ESC` toets. Houd de `CONTROL` toets ingedrukt en druk op de toets waar de `z` op staat. Zet de volgende regel nog onder je programma.

```
Typ 2 5 PRINT " zoem "
```

Je ziet nu niet het woord "zoem" op het scherm staan, maar weer een pijltje. Nu zwart-op-wit.

Vergeet niet de aanhalingstekens in de `PRINT` instructie te zetten.

```
Typ RUN
```

Als je alles goed hebt gedaan, wordt het scherm schoon gemaakt. Je ziet de zin
 IK STA OP EEN SCHOON SCHERM
 en je hoort een zoemtoon. Klopt het? De computer schrijft er `READY` onder. Hij is klaar voor de volgende opdracht. Maak het geheugen weer leeg en het scherm schoon.

Let op! Als je kleine letters ziet staan tussen de aanhalingstekens van een `PRINT` instructie, betekent dat dus altijd iets speciaals. Bijvoorbeeld:



```
1 0 PRINT "zwart-op-wit"
```

is niet

```
1 0 PRINT "ZWART - OP - WIT"
```

Typ 1 0 PRINT "

Druk op de zwart-op-wit toets.

Typ W I T

Druk weer op de zwart-op-wit toets.

Typ "

```
R U N
```

► Vaste tekst

Letters, cijfers en leestekens noemen we allemaal tekens. Zelfs een spatie is een teken. Allemaal tekens achter elkaar geschreven maken samen een 'tekst'. Een tekst tussen aanhalingstekens noemen we een 'vaste tekst'. Het is een 'tekst', omdat het bestaat uit letters, cijfers en leestekens. We noemen het een 'vaste tekst', omdat de tekens hetzelfde blijven. Een vaste tekst verandert niet wanneer het programma wordt uitgevoerd. Kijk eens naar de volgende PRINT instructies, ze hebben allemaal een 'vaste tekst':

```
Typ 1 0 PRINT " SUSAN "  
2 0 PRINT " 3 + 7 "  
3 0 PRINT " 2 8 4 "  
4 0 PRINT " IK BEN 12 JAAR "  
R U N
```

► **Opdrachten**

- 4 Maak een programma dat je voornaam, je achternaam en je adres op het scherm schrijft.
 - Je voornaam in gewone letters.
 - Een lege regel.
 - Je achternaam in zwart-op-wit letters.
 - Nog een lege regel.
 - Je adres weer in gewone letters.

- 5 Maak 3 zoemtonen in je programma:
 - De eerste voor je voornaam.
 - De tweede na je achternaam.
 - De derde na je adres.

4

Een programma veranderen

-
- De LIST instructie
 - Een regel ertussen zetten
 - Een regel uitgummen
 - Een regel veranderen
-

► **De LIST instructie**

Maak het geheugen leeg en het scherm schoon.

Typ

```
1 0 REM PROBEER PROGRAMMA
2 0 PRINT " clear"
3 0 PRINT " IK TYP DIT PROGRAMMA "
4 0 PRINT " OP MIJN "
6 0 PRINT " COMPUTER "
RUN
```

De computer heeft het programma uitgevoerd. De programmaregels staan niet meer op het scherm, maar zitten nog wel in het geheugen van de computer. Wil je regel 30 nog eens zien?

Typ LIST 30

en druk op RETURN. De computer schrijft nu alleen regel 30 op het scherm. Wil je meer regels zien?

Typ LIST 10,30

en druk op RETURN. Nu staan de regels 10, 20 en 30 op het scherm.

Dus: eerst geef je de computer de instructie LIST. Daarachter typ je het nummer van de eerste regel die je wilt

zien. Dan een komma en het nummer van de laatste regel die je wilt zien. En daarna druk je op RETURN. Je wilt het hele programma nog eens zien?

Typ LIST

(zonder regelnummer) en druk op RETURN. Je hele programma staat nu weer op het scherm.

LIST betekent: maak een lijst van de programmaregels en schrijf die op het scherm.

Wil je het programma nog eens zien zonder regelnummers en instructies?

Typ RUN

Je ziet dat de computer heeft gedaan wat er in je programma stond.

► **Een regel ertussen zetten**

Heb je een regel vergeten in je programma? Die kun je er gemakkelijk tussen zetten. Geef maar eens de instructie LIST en druk op RETURN. Het vorige programma staat nu weer op het scherm.

Typ nu de volgende regel:

Typ 50 PRINT "ATARI"

(Denk aan RETURN.) Geef nu nog eens de instructie LIST. Regel 50 staat nu netjes op de goede plaats in je programma.

► **Een regel uitgummen**

Wil je een regel uit je programma 'uitgummen'? Dat is nog gemakkelijker.

Staat je programma nog op het scherm?

Typ 6 0

zonder iets erachter en druk op `RETURN`. Geef opnieuw de instructie `LIST` en je ziet dat regel 60 is verdwenen.

► Een regel veranderen

Wil je een regel veranderen? Bijvoorbeeld regel 30? Zet dan deze regel 30 onder je programma.

```
30 PRINT " IK LEER COMPUTEREN "
```

De oude regel 30 staat nog steeds op het scherm. Geef de computer de instructie `LIST` en je ziet dat regel 30 is veranderd. Geef nu de instructie `RUN` om het nieuwe programma op het scherm te zien.

► Opdrachten

- 6 Maak zelf een programma van 6 regels.
 - Gebruik een `REM` regel.
 - Gebruik 5 `PRINT` instructies om je naam, je lievelingseten, iets dat je zelf leuk vindt om te doen, je lievelingsdier en waar je naar toe gaat met vakantie op het scherm te zetten.
 - Laat de computer je programma uitvoeren.
- 7 Breng het programma terug op het scherm.
 - Verander nu een regel en `LIST`.
 - Gum een regel uit en `LIST`.
 - Laat de computer je programma nog eens uitvoeren.

5

Hoe je iets in het geheugen stopt

-
- Variabele tekst
 - De DIM instructie
 - De doos vullen
 - De LÉT instructie
 - Dozen reserveren
 - Naam en waarde
 - De INPUT instructie
-

Weet je nog wat een ‘vaste tekst’ is? Dat is een aantal tekens tussen aanhalingstekens.

Voorbeeld 10 PRINT "KUIFJE"

► Variabele tekst

We kennen ook een ‘variabele tekst’.
Variabel betekent wisselend en dat is precies wat het is.
De tekst kan veranderen. Je kunt verschillende teksten in een doos in het geheugen stoppen.

Let op! Er kan maar één tekst tegelijk in de doos!

Een ‘variabele tekst’ krijgt altijd een naam. Dat kan een woord zijn of een willekeurige letter uit het alfabet. Maar je moet er wel altijd het teken \$ achter zetten. Zonder spatie.

De \$ is het ‘string’-teken. Bijvoorbeeld:

A\$ spreek je uit als A string.

B\$ spreek je uit als B string.

NAAM\$ spreek je uit als NAAM string.

Als je een variabele tekst in een doos in het geheugen van de computer wilt stoppen, moet je de computer eerst vertellen hoe groot die doos moet zijn. Dat doe je met de DIM instructie. Bijvoorbeeld:

```
20 DIM A$(10)
```

Met de DIM instructie geef je de computer opdracht een doos in het geheugen te reserveren voor de variabele tekst A\$. Deze doos moet groot genoeg zijn om er een tekst van 10 letters in te stoppen.

Let op! De doos wordt door de DIM instructie níét gevuld!
 ■ DIM is de afkorting van het Engelse woord dimension. Dimension betekent: grootte of afmeting.

► **De doos vullen**

Er zijn twee manieren om iets in de doos te stoppen die je met de DIM instructie hebt gereserveerd. Een ervan is de LET instructie. Met de LET instructie stop je dingen in dozen.

```
Typ 10 REM TEKST DOOS
     15 PRINT " clear"
     20 DIM A$(10)
     30 LET A$=" KUIFJE "
     40 PRINT A$
     RUN
```

De computer doet dit:

- Regel 15 – De computer maakt het scherm schoon.
- Regel 20 – De computer reserveert een doos in het geheugen. Deze doos is groot genoeg om er 10 letters in te stoppen.
- Regel 30 – De computer pakt de ‘tekst’ KUIFJE en stopt die in de doos waar A\$ op staat. De doos is gemaakt voor

10 letters. Dus past KUIFJE er gemakkelijk in, want KUIFJE heeft maar 6 letters.

- Regel 40 – De computer ziet dat hij op het scherm moet schrijven (PRINT) wat er in doos A\$ zit. Hij neemt de doos A\$ en maakt een kopie van de tekst KUIFJE, want die zit er in.

Hij zet de kopie op het scherm. De tekst KUIFJE zit nog steeds in de doos A\$.

LET A\$ = "KUIFJE" betekent: laat A\$ KUIFJE zijn.

► Meer dozen tegelijk reserveren

Met de DIM instructie kun je meer dozen tegelijk reserveren. Kijk maar:

Voorbeeld 10 DIM A\$(40),K\$(40),Z\$(40)

Regel 10 heeft 3 dozen voor variabele teksten gereserveerd met de namen A\$, K\$ en Z\$.

► Naam en waarde

In het vorige programma is de naam van de variabele tekst A\$. Deze naam wordt als het ware op de voorkant van de doos gezet.

De waarde van de variabele tekst is KUIFJE. De waarde wordt in de doos gestopt.

- Let op!** Je moet een doos reserveren met de DIM instructie voordat je iets in de doos stopt.
- Iedere doos heeft zijn eigen naam.
 - Voor de naam van de doos mag je elke letter, of een woord kiezen.
 - Je kunt net zoveel dozen reserveren als je wilt, maar elke doos moet een andere naam hebben.
 - Je mag voor elke naam maar één keer de DIM instructie gebruiken. Daarna mag je hem onbeperkt gebruiken.



► De INPUT instructie

De tweede manier om een doos te vullen is met de INPUT instructie.

INPUT betekent: stop er iets in.

Typ het volgende programma. Denk erom dat je de spaties na BOTERHAM en MET typt!

```
Typ 10 PRINT "clear "  
20 DIM A$(40),K$(40),Z$(40)  
30 LET A$=" DUS JE EET EEN BOTERHAM "  
40 LET K$=" MET "  
50 PRINT "WAT HEB JE OP JE BOTERHAM? "  
60 PRINT  
70 INPUT Z$  
80 PRINT  
90 PRINT A$;K$;Z$
```

Als je het vraagteken ziet, mag je daar iets achter typen en druk je op de RETURN toets. Als je bijvoorbeeld KAAS typt, stopt de computer dit woord in de doos met de naam z\$. Door de INPUT instructie stelt de computer je een vraag. Zag je het vraagteken?

Haal het programma terug met de LIST instructie. Dit gebeurt er:

- Regel 10 – De computer maakt het scherm schoon.
- Regel 20 – De computer reserveert 3 dozen in het geheugen, met de namen A\$, K\$ en Z\$. Elke doos is groot genoeg om er 40 letters in te stoppen.
- Regel 30 – De computer neemt de tekst DUS JE EET EEN BOTERHAM en stopt die in de doos met de naam A\$.
- Regel 40 – De computer neemt de tekst MET en stopt die in de doos B\$.
- Regel 50 – De computer schrijft WAT HEB JE OP JE BOTERHAM? op het scherm.
- Regel 60 – De computer schrijft een lege regel.
- Regel 70 – De computer zet de loper met een vraagteken op het scherm. Dit betekent dat de computer wil dat je er iets achter typt. Het woord dat je typt wordt de tekst

voor z\$. De computer pakt je woord en stopt dit in de doos met de naam z\$. Dit woord mag elke keer dat je het programma uitvoert iets anders zijn.

- Regel 80 – Dit is weer een lege regel.
- Regel 90 – De computer schrijft de waarde van A\$, K\$ en z\$ op het scherm. In deze regel staat een puntkomma. De betekenis hiervan leggen we je later uit. Telkens als je de instructie RUN typt kun je een ander antwoord geven.
- Nog een voorbeeld:

Typ

```

10 REM DRANKORGEL
15 DIM B$(30),C$(40)
20 LET B$="WAT WIL JE DRINKEN?"
25 PRINT "clear"
30 PRINT B$
35 INPUT C$
40 PRINT
45 PRINT "OH, WIL JE"
50 PRINT C$
55 PRINT "LEKKER ZEG, KRIJG IK OOK"
60 PRINT C$;"?"

```

► Opdrachten

- 8 Leg regel voor regel uit wat er in het vorige programma gebeurt.
- 9 Schrijf zelf een programma. Bijvoorbeeld over je favoriete tv-programma.
- Gebruik hierin de instructies DIM, LET en INPUT. Om het programma steeds anders te maken kun je regels uitgummen, veranderen of erbij maken. Weet je nog hoe dat ging?

6

Springen en stoppen



-
- Een sprong in je programma maken
 - De GOTO instructie
 - Een lus
 - Onder elkaar of naast elkaar
 - Naast elkaar of uit elkaar
-

► Een sprong in je programma maken

Typ

```
10 REM OP HOL GESLAGEN
15 PRINT "clear "
20 PRINT "MIJN REGEL SLAAT OP HOL "
30 GOTO 20
```

Dit programma stopt niet meer uit zichzelf. Druk op de **BREAK** toets om het programma te stoppen.

BREAK betekent: breek af of stop.

Als je op de **BREAK** toets drukt, stopt je programma meteen, waar hij ook is. Op het scherm zie je dan, bij welke regel hij is opgehouden.

```
STOPPED AT LINE XX
```

In regel 30 staat een **GOTO** instructie.

GOTO betekent: ga naar.

Iedere keer als de computer bij regel 30 komt, wordt hij teruggestuurd naar regel 20. Hij moet telkens opnieuw de **PRINT** instructie op het scherm schrijven. Dit is een oneindige 'lus'. Hij stopt niet meer. De **GOTO** instructie zullen we nog in veel programma's tegenkomen.



► **Onder elkaar of naast elkaar**

```
Typ 10 REM SPELLETJE
     20 PRINT
     30 PRINT "BOTER"
     35 PRINT "KAAS"
     40 PRINT "EN"
     45 PRINT "EIEREN"
```

Elke PRINT instructie wordt op een aparte regel geschreven.

Zet nu de volgende regels er nog onder. Denk erom dat je de spatie en de puntkomma achter BOTER, KAAS en achter EN niet vergeet!

```
Typ 60 PRINT "BOTER" ;
     65 PRINT "KAAS" ;
     70 PRINT "EN" ;
     80 PRINT "EIEREN"
```

Zie je het verschil?

Let op! De puntkomma zorgt ervoor, dat de volgende PRINT instructie op dezelfde regel wordt gezet. Let wel op de spaties! Anders worden de woorden aan elkaar geschreven.
Nog een voorbeeld. (Denk aan de spatie voor en na EN in regel 55.)

```
Typ 10 REM BEROEMD DUO
     15 DIM M$(20), P$(20)
     20 PRINT "HOE HEET JE?"
     25 INPUT M$
     30 PRINT "HOE HEET JE VRIEND(IN)?"
     35 INPUT P$
     40 PRINT "clear"
     45 PRINT "WIJ PRESENTEREN U HET BEROEM
DE DUO"
     50 PRINT
     55 PRINT M$; " EN "; P$
```



► **Naast elkaar of uit elkaar**

Typ 10 PRINT "POP";" OF "; "PUNK"
20 PRINT "POP", "OF", "PUNK"

Let op! Als je een komma zet tussen twee dingen in een PRINT instructie, haalt de computer de woorden uit elkaar en zet ze in kolommen.

► **Opdrachten**

- 10 Maak een programma dat de naam van je favoriete popgroep vraagt en één van hun liedjes.
 - Geef een PRINT instructie voor het woord ZINGT.
 - Dan geef je een PRINT instructie die de naam van de groep en de titel van het liedjes afdruckt met het woord ZINGT ertussen.
- 11 Verander het vorige programma.
 - Gebruik nu 3 PRINT instructies.
 - Zorg ervoor dat de computer deze instructies achter elkaar op het scherm schrijft.
- 12 Maak een programma dat steeds de naam van de favoriete popgroep op het scherm schrijft. Stop het programma.

7

De IF...THEN instructie



-
- De weg splitst zich
 - Gelijk aan... en niet gelijk aan...
-

IF...THEN betekent: als...dan.

In het volgende programma geven we een voorbeeld van de IF...THEN instructie.

Typ

```
10 REM VOORBEELD
15 PRINT "clear"
20 DIM L$(3)
25 PRINT "HOUD JE VAN DIEREN? (JA OF NEE)"
30 INPUT L$
35 IF L$="JA" THEN PRINT "LEUK, IK OOK"
40 IF L$="NEE" THEN PRINT "JAMMER, IK WEL"
```

Probeer de antwoorden JA, NEE en SOMS.

Wat gebeurt er:

- Regel 35 – De computer kijkt of (IF) L\$="JA". Als (IF) L\$ gelijk is aan "JA", dan (THEN) voert de computer de instructie uit die na THEN staat. Als L\$ niet gelijk is aan "JA", gaat de computer naar de volgende regel. Hij voert de instructie die na THEN staat niet uit.
- Regel 40 – De computer kijkt of L\$="NEE". Als L\$="NEE", dan voert de computer de instructie uit die na THEN staat. Dus de computer schrijft op het scherm:

```
JAMMER, IK WEL
```

Als L\$ niet gelijk is aan "NEE", gaat de computer naar de volgende regel (als die er is) en kijkt niet naar de instructie die na THEN staat.

Nog een voorbeeld:

Typ

```

10 REM ALS, ALS
15 DIM K$(15), O$(15), P$(15), R$(15)
20 LET K$="KLAAR"
25 LET O$="VOLGENDE"
30 LET P$="OPNIEUW"
35 PRINT "clear"
45 PRINT "TYP EEN WOORD"
50 INPUT R$
55 IF R$="GOED" THEN PRINT "JA, JA"
60 IF R$=K$ THEN PRINT P$
65 IF R$="VOGEL" THEN PRINT "KOEKOEK"
70 IF R$=P$ THEN PRINT O$

```

Probeer de volgende woorden: GOED, KLAAR, VOGEL en OPNIEUW. Kijk of de IF...THEN instructie werkt zoals je verwacht.

► De weg splitst zich

Als de computer een IF...THEN instructie tegenkomt, staat hij op een wegsplitsing. Hij moet kiezen welke weg hij moet nemen. Als we dit met pijlen tekenen ziet het er zo uit:

```

10 DIM F$(10)
20 PRINT "HOE VOEL JE JE?"
30 INPUT F$
40 IF F$="MOE" THEN PRINT "GA SLAPEN"
50 (nieuwe instructie)

```

Als F\$="MOE" kijkt de computer wat er na THEN staat.
Hij voert de instructie uit. Als F\$ iets anders is, gaat de
computer naar de volgende regel. Hij slaat alles over wat
er na THEN staat.

► **Gelijk aan... en niet gelijk aan...**

Het teken voor 'is gelijk aan' ken je: =

Het teken voor 'is niet gelijk aan' is < >

Om dit teken te maken druk je eerst de < toets in en dan
de > toets.

In het volgende programma staat een voorbeeld.

```
Typ  10 DIM E$(15)
      15 PRINT "HEB JE DORST? (JA OF NEE)"
      20 INPUT E$
      25 IF E$="JA" THEN PRINT "GA DAN EERST
      IETS DRINKEN"
      30 IF E$<>"JA" THEN PRINT "GA DAN VERD
      ER MET DE LES"
```

Dit gebeurt er:

- Regel 25 – Als de 'tekst' in doos E\$ gelijk is aan "JA",
schrijft de computer:
GA DAN EERST IETS DRINKEN
- Regel 30 – Als de waarde van E\$ niet gelijk is aan de
tekst "JA", schrijft de computer:
GA DAN VERDER MET DE LES

► **Opdrachten**

- 13 Maak eens een BROODJE MET programma.
- Vraag wat iemand op zijn broodje wil hebben.
 - Laat de computer iets geks antwoorden voor elk soort
beleg wat gekozen wordt. Als je bijvoorbeeld KAAS kiest,
kun je antwoorden: VANDAAG BEN JE DE BAAS.

- 14 Maak een programma RAAD DE KLEUR.
- Eén speler stopt een kleur in tekstdoos c\$.
 - De andere speler probeert de kleur te raden door achter het INPUT ? een kleur in te vullen.
 - Gebruik hiervoor 2 IF...THEN regels.
Eén regel met 'is gelijk aan', voor als het gokje goed is.
Eén regel met 'is niet gelijk aan' voor als het gokje fout is.

8

-
- Vertragen
 - Muziek
 - De SOUND instructie
 - Eenstemmig
 - De stem
 - De toonhoogte
 - De vervorming
 - De geluidssterkte
 - Tweestemmig
-

► Vertragen

Om stukken of een stuk in je programma langzamer te laten lopen gebruik je de FOR...NEXT instructie. Voorbeeld:

```
Typ 10 REM LANGZAMER
    15 PRINT " clear"
    20 PRINT " EVEN WACHTEN "
    25 FOR T=1 TO 1500:NEXT T
    30 PRINT " KLAAR "
```

In regel 25 staat: Neem voor T (ijd) 1 TO (tot) 1500: NEXT T (dan de volgende T).

Dit betekent dat de computer telt van 1 tot 1500, voor hij met de volgende instructie verder gaat. Het is net als wanneer je hem bent met verstopperje spelen.

De FOR...NEXT instructie maakt de 'vertragingsslus'.

Verander het getal 1500 in regel 25 maar eens in bijvoorbeeld 3500. Geef nog eens de instructie RUN. Je ziet dat

het nu langer duurt. Elke 500 is ongeveer 1 seconde. De FOR...NEXT instructie gebruiken we bijvoorbeeld bij het maken van muziekprogramma's.

► **Muziek**

De ATARI heeft 4 stemmen. Alle stemmen zijn hetzelfde, maar je kunt er meerdere tegelijk gebruiken. Je kunt bijvoorbeeld 2-, 3- of 4-stemmig 'zingen'.

Om geluid te maken gebruiken we de SOUND instructie. Achter de SOUND instructie moeten altijd 4 getallen staan.

Voorbeeld SOUND 1, 162, 10, 6
 SOUND S, T, V, G

- s = stem – je kunt kiezen 0, 1, 2 of 3
- T = toonhoogte – een getal van 1 t/m 255
- v = vervorming – een even getal van 0 t/m 14
- G = geluidssterkte – een getal van 0 t/m 15

Typ

```

10 REM EENSTEMMIG
15 PRINT "STEM 1"
20 PRINT "KIES EEN NOOT (1 TOT 255)"
25 INPUT T
30 PRINT "WELKE VERVORMING? (EVEN GETAL TUSSEN 0 EN 14)"
35 INPUT V
40 PRINT "HOE HARD? (0 TOT 15)"
45 INPUT G
50 SOUND 1, T, V, G
55 FOR T=1 TO 1000: NEXT T
60 GOTO 20
  
```

- De stem – 1e getal
 Het eerste getal is de stem. Je kunt kiezen uit 0, 1, 2 of 3.

- De toonhoogte – 2^e getal
Met het 2^e getal kun je de noten kiezen.
Hoe groter het getal, hoe lager de noot. Hieronder staat een aantal noten met het getal dat daarbij hoort:

Noot	Getal	Noot	Getal
C (laag)	251	C (middelste)	126
D	217	D	108
E	193	E	96
F	182	F	92
G	162	G	81
A	144	A	72
B	128	B	64
		C (hoog)	63

Heel hoge noten krijg je als je getallen in de buurt van de 10 gebruikt. Het kan zelfs zijn dat je de nummers 2, 1 en 0 niet meer kunt horen.

- De vervorming – 3^e getal
Je kunt de 'kwaliteit' van het geluid verbeteren door het getal dat de vervorming aangeeft te veranderen. Het getal 10 geeft de zuiverste tonen.
Voor speciale effecten is het juist leuk om andere getallen te nemen. Je kunt het geluid van onweer, het gesis van een slang, een geweer of een ontploffing namaken. Gewoon proberen.
- De geluidssterkte – 4^e getal
Het getal 0 zet het geluid af. De gewone geluidssterkte is het getal 8. Het hardst is 15. Natuurlijk kun je ook je tv harder zetten.

Let op! Als je meer stemmen tegelijk gebruikt mogen de getallen van de geluidssterkte bij elkaar opgeteld nooit meer dan 32 zijn!

Voorbeeld

```

15 SOUND 1,180,10,8
18 SOUND 2,42,6,12
20 SOUND 3,240,12,10

```

Dit is goed, want $8 + 12 + 10$ is 30.

Typ

```

10 REM TWEESTEMMIG
15 PRINT "DE STEMMEN 0 EN 1"
20 PRINT "KIES EEN NOOT VOOR STEM 0"
25 INPUT T0
30 PRINT "KIES EEN NOOT VOOR STEM 1"
35 INPUT T1
40 SOUND 0,T0,10,8
45 SOUND 1,T1,10,8
50 FOR T=1 TO 850:NEXT T
60 GOTO 20

```

► Opdrachten

- 15 Maak een lijstje van geluiden met diverse vervormingsgetallen.
 - Schrijf op waar het geluid op lijkt.
 - Bedenk een spelletje waarin je het geluid, of het effect kunt gebruiken. Als dit nog te moeilijk is probeer het dan eerst met één vervormingsgetal.
- 16 Probeer eens of je het geluid van een politiesirene kunt maken. Hiervoor gebruik je 2 SOUND instructies voor één stem. Om de tonen langer of korter te laten duren, gebruik je de FOR...NEXT instructie.

9

Werken met getallen



-
- Geheugendozen
 - Variabelen
 - De reken-tekens
 - Rekenen met de LET instructie
 - Het = teken liegt soms
-

Tot nu toe heb je meestal een 'tekst' gebruikt bij de instructies PRINT, LET en INPUT. Je gaat ze nu ook voor getallen gebruiken.

Typ het volgende programma:

```
Typ 10 REM GROTER
     15 PRINT " clear"
     20 PRINT "GEEF ME EEN GETAL"
     25 PRINT "VAN TEN HOOGSTE 5 CIJFERS"
     30 INPUT G
     35 LET A=G+1
     40 PRINT
     45 PRINT "DE COMPUTER GEEFT JOU"
     50 PRINT "EEN GROTER GETAL TERUG"
     55 PRINT A
     60 GOTO 20
```

Wat gebeurt er in het programma?

- Regel 30 – De computer wacht netjes tot je een getal intoetst. Hij neemt het getal en gaat op zoek naar de doos met de naam G. De computer kan deze doos niet vinden, want we hebben G nog niet eerder in het programma gebruikt. Hij pakt dus een nieuwe doos, zet de naam G erop, en stopt het getal erin.

- Regel 35 – De computer neemt het getal uit doos G. Hij telt er 1 bij op. Hij stopt dit nieuwe getal in een andere doos, en zet er de naam A op.
- Regel 55 – De computer neemt de doos met A erop. Hij pakt een kopie van het getal dat erin zit, en schrijft dit getal op het scherm. Het originele getal A zit nog steeds in de doos.
- Regel 60 – De computer gaat terug naar regel 20, zodat je nog eens een getal kunt geven.

► Geheugendozen

Net als een ‘tekst’ worden ook getallen in een doos gestopt. Er is één verschil:

Let op! TEKST: Je moet een DIM instructie geven om de computer te vertellen hoe groot de doos moet zijn waar je de tekst in wilt stoppen.

GETAL: Je gebruikt géén DIM instructie. Alle dozen waar je een getal in stopt zijn even groot.

► Variabelen

Achter de naam van de doos waar je een ‘variabele tekst’ in stopt staat altijd dit teken: \$. Bijvoorbeeld: A\$, K\$, N\$. De naam van de doos waar een ‘variabel getal’ in zit heeft dit niet. Je laat het \$ teken dan weg. Bijvoorbeeld: A, K, N.

► De reken-tekens

Het – teken, het = teken, het + teken en het × teken zitten op de pijltoetsen. De computer gebruikt een sterretje als vermenigvuldigingsteken. Het deel-teken maak je zó: /. Het zit op dezelfde toets als het vraagteken.



► Rekenen met de LET instructie

```

Typ 10 LET A = 8
     15 LET B = 12
     20 LET C = A + B
     25 PRINT " C = " ; C
     30 LET D = C * A
     35 PRINT " D = " ; D

```

In de LET instructie mag je het woord LET weglaten. De computer weet dat je LET bedoelt als je de regel begint met de naam van de 'variabele', gevolgd door het = teken. Deze beide regels doen precies hetzelfde:

```

10 LET R = 12           10 R = 12

```

en deze regels ook:

```

10 LET A$ = " HOI "    10 A$ = " HOI "

```

In het volgende programma hebben we LET weggelaten.

```

Typ 10 A = 35
     20 B = A / 7
     30 Z = B * ( 3 + 15 )
     40 PRINT " B = " ; B
     50 PRINT " Z = " ; Z

```

Let op! Een getal tussen aanhalingstekens is een "tekst". "1984" is geen getal, maar een tekst, want het staat tussen aanhalingstekens. Je kunt niet rekenen met cijfers die in een tekst staat.

```

Goed: 10 LET A = 2 + 4      Fout: 10 LET A$ = 2 + 4
Goed: 20 B = 12 - 5        Fout: 20 B = " 12 " - " 5 "

```

Als je zo'n foute regel typt zegt de computer ERROR-. Kijk maar eens goed naar deze regels. Welke regel is er fout en waarom? Kijk maar eens wat de computer doet.

Typ 1 0 DIM A\$=12
 2 0 LET E=8
 3 0 LET A\$="10"
 4 0 LET K=E+A\$

Let op! Je kunt getallen en teksten wel in dezelfde PRINT instructie zetten, maar je kunt er dan níét mee rekenen. Kijk maar:

Typ 1 0 PRINT 4*2;" IS GELIJK AAN 4*2"

► **Het = teken liegt soms**

Het = teken betekent bij het computeren niet altijd 'is gelijk aan'. Bekijk deze regel:

2 0 A = A + 1

Rekenkundig kan dat natuurlijk niet. Stel dat $A = 5$. Dan staat er: $5 = 5 + 1$ dus $5 = 6$. Dat klopt niet. Maar bij computeren mag je wél zeggen $A = A + 1$. Het = teken betekent dan: vervang dit door. Wat gebeurt er met deze regel:

2 0 A = A + 1

- De computer gaat naar de doos met de naam A erop. Hij pakt het getal 5 uit de doos en telt er 1 bij op. Dan stopt hij het nieuwe getal 6 weer in de doos. Het betekent dus eigenlijk: de nieuwe A is gelijk aan de oude $A + 1$. We zullen dit in raadspelletjes veel gebruiken.

► **Opdrachten**

- 17 Maak een programma dat twee getallen vraagt.
- De getallen worden dan vermenigvuldigd en op het scherm geschreven.



- 18 Maak een programma dat naar je leeftijd vraagt, en naar het jaar waarin we nu leven.
- Trek deze beide getallen van elkaar af en laat de computer je geboortjaar als antwoord geven. Word je nog jarig dit jaar? Vul dan in hoe oud je wordt. Anders geeft de computer een verkeerd jaartal.
 - Denk eraan dat je PRINT instructies gebruikt. Hierin zet je wat er gevraagd wordt. In de laatste PRINT instructie zet je bijvoorbeeld: " JE BENT GEBOREN IN " .

10

Getallen in de IF...THEN instructie



- Nog meer rekenkundige tekens
- WAAR OF NIET WAAR
- De dubbele punt
- Fouten met de dubbele punt

► Nog meer rekenkundige tekens

In de rekenkunde kennen we nog meer tekens dan we in les 9 hebben gehad. Kijk maar:

Voorbeeld Gelijk aan =
 Kleiner dan <
 Groter dan >
 Ongelijk aan <>

Als er staat:

3 < 4 zeggen we 3 is kleiner dan 4

3 <> 4 zeggen we 3 is ongelijk aan 4

4 > 3 zeggen we 4 is groter dan 3

In het volgende programma staan een paar voorbeelden.

```
Typ 10 REM HOEVEEL EET JE
     15 PRINT "clear"
     20 PRINT "HOEVEEL BOTERHAMMEN EET JE?"
     25 INPUT B
     30 IF B<4 THEN PRINT "JE BENT OOK GEEN
        GROTE ETER!"
     35 IF B>5 THEN PRINT "TJEE, JIJ LIJKT
        HOLLE BOLLE GIJS WEL"
```



```

40 IF B=4 THEN PRINT "NOU DAT LIJKT ME
HEEL AARDIG"
50 IF B=5 THEN PRINT "DA'S EEN HANDVOL
BROOD"

```

► **WAAR of NIET WAAR**

De volgende stellingen zijn WAAR of NIET WAAR.
Zeg welke stelling WAAR is en welke NIET WAAR.

Stel $A = 8$	$B = 4$	$C = 4$
$A = B$	WAAR	NIET WAAR
$A > B$	WAAR	NIET WAAR
$A < C$	WAAR	NIET WAAR
$B = C$	WAAR	NIET WAAR
$B < > C$	WAAR	NIET WAAR

Een stelling of gezegde kan dus WAAR of NIET WAAR zijn.
Kijk maar:

sneeuw is wit	WAAR	NIET WAAR
rood is blauw	WAAR	NIET WAAR
water is nat	WAAR	NIET WAAR
ijs is lekker	WAAR	NIET WAAR

Je ziet, soms is een stelling voor de ene persoon WAAR,
maar voor een ander NIET WAAR.

Ik vind ijs lekker, voor mij is dat dus WAAR, maar ik weet
niet of jij ijs ook lekker vindt.

Hier is nog een programma:

```

Typ 10 LET B=1
15 PRINT "clear"
20 IF B=13 THEN PRINT "PECHVOGEL"
25 FOR T=1 TO 750: NEXT T
30 LET B=B+2
40 IF B>25 THEN GOTO 80
50 GOTO 20
60 PRINT
80 PRINT "KLAAR"

```



Dit gebeurt er in het programma:

- Regel 10 – De computer stopt het getal 1 in geheugendoos B.
- Regel 15 – Maakt het scherm schoon.
- Regel 20 – Als de stelling $B=13$ WAAR is, schrijft de computer PECHVOGEL op het scherm.
- Regel 30 – De computer gaat naar geheugendoos B. Hij haalt het getal 1 eruit en telt er 2 bij op. Dan stopt hij het nieuwe getal terug in de doos.
- Regel 40 – Als de stelling 'B is groter dan 25' WAAR is, gaat de computer naar regel 80. Hij voert dan de instructie uit die daar staat.
Als de stelling 'B is groter dan 25' NIET WAAR is, gaat de computer naar de volgende regel.
- Regel 50 – De computer gaat terug naar regel 20. Dat doet hij zolang tot hij in doos B een getal vindt dat groter is dan 25.
- Regel 60 – PRINT instructie voor een lege regel.
- Regel 80 – PRINT instructie voor het woord KLAAR.

Hoe vaak denk je dat de computer terug moet gaan naar regel 20, voor hij het getal 25 in doos B vindt?

Wat komt er op het scherm te staan als je regel 10 verandert in `LET B=2`?

► De dubbele punt

Je kunt een dubbele punt gebruiken om meerdere opdrachten in dezelfde regel te zetten. Dit bespaart geheugenruimte.

Voorbeeld Zonder:

```
10 IF X<>2 THEN GOTO 90
15 A=2
20 B=5
25 PRINT
```



```
30 PRINT "GOED ZO"
90 (nieuwe instructie)
```

Met:

```
10 IF X=2 THEN A=2 : B=5 : PRINT : PRINT "GOED ZO"
90 (nieuwe instructie)
```

► **Fouten met de dubbele punt**

Maar... let op!

Voorbeeld

```
10 IF X=N THEN PRINT "GEWONNEN"
15 L=M
```

Is niet hetzelfde als:

```
50 IF X=N THEN PRINT "GEWONNEN" : L=M
```

Want L=M in regel 15 wordt altijd uitgevoerd, ongeacht of X=N WAAR of NIET WAAR is. L=M in regel 50 wordt alleen uitgevoerd als X=N WAAR is.

Een REM en een GOTO instructie moeten altijd aan het eind van een regel worden gezet. Alles wat na een REM of een GOTO staat, slaat de computer over.

```
Goed: 25 L=25 : REM L IS DE LENGTE
Fout: 25 REM L IS DE LENGTE : L=25
```

Na het zien van een REM gaat de computer naar de volgende regel. Hij ziet dus niet meer L=25.

Goed:

```
60 X+B=4 : GOTO 100
70 A=B-1
```



Fout:

```
60 X+B=4 : GOTO 100 : A=B-1
```

In de foute regel 60 gaat de computer naar regel 100 en komt niet terug om $A=B-1$ uit te voeren.

Typ

```
10 REM RAAD HET GETAL
12 REM SPEL VOOR TWEE PERSONEN
20 PRINT "clear"
30 PRINT "SPELER 1 GEEFT EEN GETAL"
40 INPUT A
50 PRINT "clear"
60 PRINT "RAAD HET GETAL"
70 INPUT B
80 IF B<A THEN PRINT "TE KLEIN"
90 IF B>A THEN PRINT "TE GROOT"
100 IF B=A THEN GOTO 120
110 GOTO 60
120 PRINT "GOED GERADEN"
```

► Opdrachten

- 19 Maak het vorige programma zo, dat er meerdere instructies in dezelfde regel komen te staan.
- 20 Maak zelf een RAAD HET WOORD spelletje. De tweede speler mag vier keer raden, dan is het spelletje afgelopen.
 - Gebruik hiervoor de instructies:

```
R=R+1
IF R=4 THEN PRINT "NIET GERADEN,
SPEL AFGELOPEN"
```

11

Toevallige getallen en de functie INT

- Het toeval, de functie RND
- De functie INT
- Wat zet je allemaal tussen ()

► Het toeval

Als je een dobbelsteen gooit, weet je niet van tevoren welk getal er boven komt te liggen. Dat hangt af van het toeval. Als je kaarten deelt kun je ook niet van tevoren voorspellen welke kaarten iemand krijgt. (Anders speel je vals.)

In sommige programma's heb je het toeval nodig. Hiervoor gebruiken we de functie RND: RND is de afkorting van het Engelse woord random.

Random betekent: toeval of 'zo maar'.

In een programma ziet dat er zo uit:

```
Typ 10 REM TOEVALLIGE GETALLEN
     15 PRINT "clear"
     20 T=RND(0)
     25 PRINT T
     30 IF T<0.95 THEN GOTO 20
```

Er staan nu een heleboel decimale getallen op het scherm. Deze zijn gemaakt door de RND functie in regel 20. Met decimaal bedoelen we getallen achter de komma. Bijvoorbeeld 0,386.

De computer zet op de plaats waar wij meestal een komma zetten een punt. Daar moet je dus even op letten.

Na het woord RND volgt altijd een getal tussen haakjes. Het geeft niet welk getal, als het maar groter dan 0 is. RND (0) geeft getallen die groter zijn dan 0, maar kleiner dan 1. Wil je getallen groter dan 1, dan hoef je alleen maar te vermenigvuldigen. Kijk maar. Verander de regels 20 en 30 uit het vorige programma:

```
Typ 20 T=RND(0)*23
     30 IF T<21 THEN GOTO 20
```

T<21 is willekeurig gekozen. De getallen op het scherm zitten nu tussen de 0 en de 23. Maar meestal hebben we liever hele getallen. Zoals 3 en 12, in plaats van 3,857329 of 12,9532856.

► **De functie INT**

Om hele getallen te krijgen gebruiken we de functie INT. INT is de afkorting van het Engelse woord integer. Integer betekent: hele.

De INT functie neemt het totale getal, bijvoorbeeld 8,36195. Hij laat alle cijfers achter de punt weg en houdt alleen de 'hele', dus 8 over. Probeer de INT functie in dit programma:

```
Typ 10 A=INT(3.9)
     15 PRINT A
```

Denk eraan een punt te typen in 3.9 in plaats van een komma!

Nog een voorbeeld:

```
Typ 10 A=3.9
     12 K=5.7
     15 Z=A+K
     18 P=INT(A+K)
     20 PRINT Z,P
```

Kijk naar de antwoorden. Je ziet dat de INT functie de cijfers achter de komma (punt) heeft weggelaten. De INT functie zie je het duidelijkst in dit programma:

```

Typ  10 REM INT FUNCTIE
        15 PRINT " clear"
        20 PRINT " GEEF ME EEN GETAL MET "
        22 PRINT " 5 CIJFERS ACHTER DE PUNT "
        30 INPUT G
        40 A=INT(G)
        50 PRINT " HET DECIMALE GETAL IS ";G
        60 PRINT " DE HELE HIERVAN IS ";A
    
```

Wil je het programma een paar keer doen? Zet deze regel er dan nog onder:

```

80 GOTO 20
    
```

► **Wat zet je allemaal tussen ()**

```

Getallen      : A=INT(42.5)
Variabelen    : A=INT(M)
Berekeningen : A=INT(4*(X+6))
Functies      : A=INT(RND(0))
    
```

We laten je nu zien hoe je een hoop regels kunt besparen:

Voorbeeld In plaats van:

```

25 A=RND(0)
30 B=A*4
35 I=INT(B)
40 Y=2+1
    
```

typ je:

```

40 Y=2+INT(RND(0)*4)
    
```

Hier is nog een programma. We gooien met twee dobbelstenen tegelijk. Op het scherm komen de getallen te staan die we hebben 'gegooid'. We gebruiken de INT en RND functies in dezelfde regel.

```

Typ 10 REM DOBBELSTENEN GOOIEN
      15 DIM B$(3)
      20 PRINT "clear"
      30 A=INT(RND(0)*7)
      35 IF A<1 THEN GOTO 30
      40 B=INT(RND(0)*7)
      45 IF B<1 THEN GOTO 40
      50 PRINT "OP DE EERSTE DOBBELSTEEN STA
      AT EEN: ";A
      60 PRINT
      70 PRINT "OP DE TWEEDE DOBBELSTEEN STA
      AT EEN: ";B
      80 PRINT:PRINT
      90 PRINT "NOG EEN KEER? JA OF NEE"
     100 INPUT B$
     110 IF B$="JA" THEN GOTO 20
  
```

► **Opdrachten**

- 21 Probeer of je regel voor regel uit kunt leggen hoe het DOBBELSTEEN programma werkt.
- 22 Maak zelf een programma waarin je 4 lottocijfers voorspelt. (Of, als je veel tijd hebt, alle.)

12

Het opslaan van gegevens

-
- Weinig gegevens in een programma zetten
 - Veel gegevens in een programma zetten
 - De wijsvinger van READ
 - Terug naar af
 - Gemengde gegevens
-

► Weinig gegevens in een programma zetten

Kleine hoeveelheden gegevens kun je op twee manieren in een programma zetten. Je gebruikt hiervoor de INPUT instructie of de LET instructie.

- Hier is het eerste voorbeeld-programma. In dit programma is P\$ het 'gegeven'. Het wordt pas ingetypt als het programma wordt uitgevoerd (RUNt).

Typ

```

10 REM INPUT VOORBEELD
15 DIM P$(40)
20 PRINT "clear"
30 PRINT "IK HEB EEN HEKEL AAN "; INPUT P$
40 PRINT "MEEN JE DAT NOU?"
50 PRINT "JE HEBT DUS EEN HEKEL AAN"
60 PRINT P$

```

- Dit is het tweede voorbeeld-programma. In dit programma zijn A en B de 'gegevens'. Je zet ze in het programma terwijl je het maakt.

```

Typ  10 REM LET VOORBEELD
      20 PRINT "clear"
      25 A=184
      30 B=596
      40 PRINT A-B
  
```

► **Veel gegevens in een programma zetten**

Als je veel gegevens in een programma moet opslaan gebruik je een DATA opdracht.

DATA betekent: gegeven(s).

Je gebruikt de READ opdracht om de gegevens uit de DATA regel te krijgen.

READ betekent: lees.

```

Typ  10 REM VEEL GEGEVENS
      15 DIM A1$(12), A2$(12), A3$(12), A4$(12)
      20 PRINT "clear"
      25 DATA ZONDAG, MAANDAG, DINSDAG, WOENSDAG,
      DONDERDAG, VRIJDAG, ZATERDAG
      30 READ A1$
      35 READ A2$
      40 READ A3$
      45 READ A4$
      50 PRINT A1$, A4$
  
```

Als je het programma uitvoert (RUNt) zit in doos A1\$ het eerste gegeven van de DATA regel: dus ZONDAG. In doos A4\$ zit het vierde gegeven: dus WOENSDAG.

- Het is onbelangrijk waar de DATA regels in een programma staan.
Verander regelnummer 25 maar in regelnummer 80.
Geef de instructie RUN.
- Je ziet, het programma werkt precies hetzelfde.
- Het geeft niet hoeveel DATA regels er in je programma staan. Doe dit: verdeel de DATA regel in twee stukken. Je zult zien dat het programma hetzelfde werkt.

Typ 80 DATA ZONDAG, MAANDAG, DINS DAG
85 DATA WOENSDAG, DONDERDAG, VRIJDAG, ZATERDAG

► **De wijsvinger van READ**

Stel je voor dat de READ instructie een lange wijsvinger zou hebben. Hij wijst daarmee altijd het volgende gegeven aan dat hij moet lezen. Als het programma begint, heeft READ zijn wijsvinger bij het eerste gegeven in de eerste DATA regel. (De eerste DATA regel heeft het laagste regelnummer van alle DATA regels in het programma.) Telkens als het programma een READ instructie heeft uitgevoerd, gaat de wijsvinger van READ naar het volgende gegeven van de DATA regel.

Als het laatste gegeven van een DATA regel is gelezen, gaat de wijsvinger van READ automatisch naar het eerste gegeven van de volgende DATA regel. Het geeft niet of er nog andere regels tussen staan.

Verander regel 80 maar weer in regel 25, en laat regel 85 staan. Geef de instructie RUN.

Je ziet dat het nog steeds hetzelfde werkt.

Let op! Als READ zijn wijsvinger bij het laatste gegeven van de laatste DATA regel heeft, zijn er geen gegevens meer om te lezen.

Probeer je toch nog een READ opdracht te geven, dan krijg je de volgende boodschap van de computer: ERROR-

► **Terug naar af**

Op elk punt in je programma heb je de keuze uit twee mogelijkheden.

- Je kunt nog een READ opdracht geven. De wijsvinger gaat dan naar het volgende gegeven in de DATA regel.
- Je kunt een RESTORE opdracht geven. Onder het programma VEEL GEGEVENS kun je daartoe de volgende regel zetten:

Je kunt nu weer een nieuwe READ opdracht geven. Deze READ begint dan weer met het eerste gegeven in de eerste DATA regel. Als het gegeven waarvoor je een READ opdracht hebt gegeven, het laatste gegeven is dat je wilt hebben, en je wilt daarna weer van voren af aan beginnen in je DATA regels moet je een RESTORE opdracht geven.

► Gemengde gegevens

In een DATA regel kun je net zoveel ‘getallen’ en ‘teksten’ zetten als je maar wilt. Maar je moet wel oppassen dat je in de READ opdracht de juiste ‘variabele’ zet om de gegevens te lezen.

Voorbeeld Goed:

```
45 DATA 128, ONZIN
50 READ A (goed: 128 is een getal)
55 READ A$ (goed: ONZIN is een tekst)
```

Fout:

```
45 DATA 128, ONZIN
50 READ A$ (fout: 128 is nu een tekst)
55 READ A (fout: ONZIN is geen getal)
```

► Opdrachten

- 23 Probeer of je een programma kunt maken met de namen van je familie. Als je aan de computer `BROER` vraagt, geeft hij je de namen van al je broers. In dit programma hebben de DATA regels paren van gegevens. Het eerste gegeven is de familie-relatie zoals vader, moeder, oom, zus enz.

Het tweede gegeven is de naam van het familielid. Je kunt natuurlijk doen alsof je een heleboel broers en zussen hebt. Elk familielid krijgt een eigen DATA regel.

- 24 Maak een ZO DENK IK OVER JOU programma.
- Het programma vraagt de naam van je onderwijzer(es) op school.
- In het programma staan ten minste drie zinnen die vertellen hoe jij over hem/haar denkt.
 - Het programma kiest willekeurig één van de zinnen en schrijft die op het scherm achter de naam van je meester of juf (het willekeurig kiezen doe je met de RND functie).

13

Kleuren



-
- Kleurdozen en kleurpotloden
 - Een gekleurd scherm
 - Gekleurde randen
-

► Kleurdozen en kleurpotloden

Met de ATARI kun je ook kleuren op het scherm brengen.

Je hebt 5 kleurdozen. Deze dozen hebben de nummers: 0, 1, 2, 3 en 4. In iedere doos zitten 16 kleurpotloden.

ATARI heeft elk kleurpotlood een nummer gegeven. Deze nummers, met de kleuren, staan hieronder:

0 = zwart-grijs-wit	8 = blauw
1 = lichtoranje (goud)	9 = lichtblauw
2 = oranje	10 = turquoise
3 = oranje-rood	11 = blauw-groen
4 = roze	12 = groen
5 = paars-rood	13 = geel-groen
6 = paars-blauw	14 = oranje-groen
7 = blauw	15 = lichtoranje

Zwart is kleur 0 met helderheid 0.

Wit is kleur 0 met helderheid 14.

Grijs zit hier tussenin.

Je kunt voor elke kleur een helderheid kiezen. Dit zijn nummers tussen 0 en 14. Maar alléén even nummers. Hoe hoger het nummer is, hoe lichter de kleur wordt.



Om de achtergrond een andere kleur te geven gebruiken we kleurdoos nummer 2.

Om te kunnen kleuren, moet je de computer de SETCOLOR instructie geven.

SETCOLOR betekent: maak een kleur.

Na het woord SETCOLOR komen altijd 3 getallen te staan, gescheiden door komma's.

1e getal: dit is het nummer van de kleurdoos. (0 t/m 4)

2e getal: dit is het nummer van het kleurpotlood. (0 t/m 15)

3e getal: dit is het nummer van de helderheid. (even nummers van 0 t/m 14)

► Een gekleurd scherm

Met dit programma kun je het tv-schermbeschrijving verschillende kleuren geven.

```
Typ 10 REM GEKLEURD SCHERM
20 PRINT "MET KLEURDOOS 2 KLEUR IK HET
SCHERM"
30 PRINT "WELKE KLEUR MOET HET SCHERM
KRIJGEN?"
35 PRINT "KIES EEN NUMMER"
36 PRINT "TUSSEN 0 EN 15"
38 INPUT K
40 PRINT "WELKE HELDERHEID?"
45 PRINT "KIES EEN EVEN NUMMER TUSSEN
0 EN 14"
46 INPUT H
50 SETCOLOR 2 , K , H
60 GOTO 30
```

Regel 50 vertelt de computer dat hij kleurdoos nummer 2 moet pakken.

De K zegt welk kleurpotlood hij moet nemen.

De H zegt hoe helder die kleur moet zijn.



Probeer eerst kleur 9 met helderheid 4. Zie je dat er niets verandert? Dit is de gewone kleur van het scherm en de letters.

Probeer nu eens kleur 9 met helderheid 14. Je ziet dat de letters nu donkerder zijn dan de achtergrond.

Met kleurdoos 2 kies je dus de kleur van de achtergrond en van de letters.

De letters hebben dezelfde kleur als de achtergrond, maar ze hebben een andere helderheid. Normaal hebben de letters helderheid 10. Dat doet de computer vanzelf.

Kijk eens wat er gebeurt als je ook helderheid 10 kiest voor de achtergrond. Juist, je ziet geen tekst meer op het scherm. De achtergrond en de letters hebben beide dezelfde helderheid gekregen.

Druk op de BREAK toets om het programma te stoppen.

Let op! Als je in de knoei zit, en je niets meer op het scherm ziet, druk dan op de RESET toets. Alles is dan weer gewoon.

► **Gekleurde randen**

Kleurdoos nummer 2 hebben we gebruikt om er het hele scherm mee te kleuren.

Kleurdoos nummer 4 gebruik je om gekleurde randen te maken.

Typ

```
1 0 REM - - - KLEURIGE RANDEN - - -
15 PRINT "MET KLEURDOOS 4 KLEUR IK DE
RANDEN"
2 0 PRINT "VOOR DE KLEUR KIES IK"
25 PRINT "EEN NUMMER TUSSEN 0 EN 15"
3 0 INPUT K
4 0 PRINT "VOOR DE HELDERHEID KIES IK"
45 PRINT "EEN EVEN NUMMER TUSSEN 0 EN
14"
5 0 INPUT H
55 SETCOLOR 4 , K , H
6 0 GOTO 2 0
```

ATARI
MUSEUM.NL

In plaats van het woord PRINT te typen mag je ook alleen een vraagteken typen. De computer weet dat een vraagteken hetzelfde betekent als PRINT. Probeer maar:

Typ 10 ? "HALLO ALLEMAAL"

► **Opdrachten**

- 25 Maak een programma dat iemands naam vraagt.
 - Laat de randen van kleur veranderen.
 - Laat de computer een gek antwoord geven aan degene die zijn naam heeft ingetypt.
- 26 Maak een programma waarin je het scherm en de randen een andere kleur geeft.
 - Probeer welke kleuren leuk bij elkaar passen.

14

Programma's bewaren op cassettebandjes

-
- Het bewaren van een programma
 - Maak een lijst
 - Het programma weer in de computer brengen
-

Leuke programma's wil je misschien bewaren om er later nog eens mee te spelen. Hieronder volgen de instructies hoe je dat moet doen.

Typ dit hele korte oefen-programma.

Typ 10 REM BEWAREN OP CASSETTE
20 ? "HALLO ALLEMAAL"

► Bewaren van een programma

Stop een cassette-bandje in de ATARI 1010 programma-recorder. Spoel het bandje helemaal terug. Druk het kleine knopje naast de teller in. Deze knop zet de teller op 000.

Als je een gewone muziek-cassette hebt, druk dan de toets ADVANCE in en spoel het bandje door tot de teller op 010 staat. Dit is niet nodig bij speciale computer-cassettes.

Typ CSAVE

en druk op de RETURN toets.

CSAVE betekent: bewaar op cassette. Je hoort nu twee keer een zoemtoon. Dat is het sein om twee toetsen van de programmarecorder in te drukken.



Druk nu de toetsen RECORD en PLAY op de recorder tegelijk in.

Druk op de RETURN toets van de computer. De computer start nu de motor van de programma-recorder. Eerst wordt er een 'aanloopstrook' gemaakt. Dit duurt ongeveer 10 seconden. Daarna begint de computer het programma op het bandje te zetten.

Als de opname klaar is, zet de computer de motor van de programma-recorder af.

Op het scherm staat nu: READY

Druk nu op de STOP toets van de recorder.

► **Maak een lijstje**

Je kunt natuurlijk de naam van het programma op de voorkant van het cassette-bandje zetten. Schrijf het getal dat de teller aanwijst als het programma begint en als het is opgenomen erachter.

Voorbeeld Programma DOBBELSTENEN GOOIEN 010-035

Er gaan veel meer programma's op een bandje. Gebruik de teller van de recorder. Zet de lengte van elk programma op een lijstje. Zo kun je meerdere programma's op een bandje zetten. Het is handig om tussen de verschillende programma's telkens 10 nummers op de teller verder te spoelen.

Let op! Als je een belangrijk programma wilt bewaren kun je er beter nog een kopie van maken. Wanneer je een keer per ongeluk je programma hebt verknoeid, heb je er nog een extra. Deze extra kopie kun je op de achterkant van het cassette-bandje zetten. Je kunt voor je extra kopieën ook een apart bandje nemen.

► **Het programma weer in de computer brengen**

We gaan eens kijken of je het programma dat je net op



een bandje hebt gezet, weer in de computer kunt krijgen.

Maak eerst het geheugen leeg en het scherm schoon. Dit doe je omdat je anders niet weet of dat, wat er op het scherm komt te staan, nog was achtergebleven van een vorig programma, of dat je het vanaf de programma-recorder hebt geladen.

Spoel het bandje terug tot de teller weer op 000 staat (of 010 voor muziek-cassettes).

Typ CLOAD

en druk op de RETURN toets.

CLOAD betekent: laad vanaf de programma-recorder.

De zoemer gaat nu één keer.

Dat betekent dat je één toets van de programma-recorder moet indrukken.

Druk de PLAY toets van de recorder in.

Druk op de RETURN toets van de computer.

De computer start nu de motor van de recorder en kijkt of er een programma op het cassette-bandje staat. Je ziet de teller van de recorder lopen.

Als het programma weer terug is in het geheugen van de computer, komt er op het scherm te staan: READY

Typ LIST om te kijken of het programma dat nu in het geheugen van de computer zit juist is.

► **Opdrachten**

- 27 Maak een programma en zet dit op een bandje.
- 28 Maak nog een programma en zet dit achter het eerste programma op hetzelfde bandje.
- 29 Breng beide programma's afzonderlijk terug in het geheugen van de computer.

Je bent nu aan het eind gekomen van dit boekje. We hopen dat je veel plezier hebt gehad in het leren omgaan met de computer.

Ben je nieuwsgierig geworden? Achterin dit boekje staan nog een paar programma's en spelletjes die je over kunt typen.

We wensen je nog veel computer-plezier.

Opdracht 6

```

10 REM!!!HET LIEFST!!!
20 DIM A$(50),B$(50),C$(50),D$(50),F$(
50)
25 PRINT "clear"
26 PRINT "HOE HEET JE?":INPUT F$
30 PRINT "WAT EET JE HET LIEFST?"
35 INPUT A$
40 PRINT "WAT DOE JE HET LIEFST?"
45 INPUT B$
50 PRINT "WAT IS JE LIEVELINGSDIER?"
55 INPUT C$
60 PRINT "WAAR GA JE HEEN OP VAKANTIE?"
65 INPUT D$
66 PRINT "clear"
70 PRINT "IK WEET EEN HELEBOEL VAN JOU"
73 PRINT:PRINT "JE HEET ";F$
75 PRINT "JE EET HET LIEFST ";A$
80 PRINT "JE LIEVELINGSDIER IS ";C$
85 PRINT "EN JE GAAT HET LIEFST ";B$;"
IN ";D$

```

Opdracht 13

```

10 REM >>>BROODJE MET<<<
15 DIM A$(20),B$(20),C$(20),D$(20)
18 DIM Z$(50)
20 LET A$="KAAS"
21 LET B$="KROKET"
22 LET C$="WORST"
23 LET D$="FRIET"

```

```

24 PRINT "clear"
25 PRINT " HALLO IK BEN MARC, DE BROODJ
ES-MAN. WAT WIL JE EROP VANDAAG?"
26 PRINT "KIES UIT KAAS, KROKET, WORST
EN FRIET. MEER HEBBEN WE NIET"
30 PRINT "ZEG HET MIJ. IK RIJM ERBIJ"
35 INPUT Z$
40 IF Z$=A$ THEN PRINT "O.K. VANDAAG B
EN JIJ DE BAAS"
45 IF Z$=B$ THEN PRINT "ZEG, MOET JIJ
NOG NIET NAAR BED"
50 IF Z$=C$ THEN PRINT "NIET DOEN JOH,
DAARVAN KRIJG JE DORST"
55 IF Z$=D$ THEN PRINT "HA HA, DAT HEB
BEN WE NIET"

```

Opdracht 16

```

5 REM >SIRENE<
10 SOUND 0,114,10,10
20 FOR T=1 TO 200:NEXT T
30 SOUND 0,85,10,10
40 FOR T=1 TO 300:NEXT T
50 GOTO 10

```

Opdracht 20

```

10 REM ++RAAD HET WOORD++
11 REM IN DIT SPEL IS HET TE RADEN WOO
RD KROKODIL
15 PRINT "clear"
18 R=0
20 DIM A$(25)
25 PRINT "RAAD HET DIER"
26 PRINT "JE MAG 4 KEER RADEN"
30 INPUT A$
31 R=R+1
32 IF A$="KROKODIL" THEN GOTO 200
35 IF R=4 THEN GOTO 210

```

```

36 IF A$<>"KROKODIL" THEN PRINT ."NIET
GOED. PROBEER HET NOG EENS"
37 PRINT
45 GOTO 30
200 PRINT "GOED ZEG!GERADEN! ":END
210 PRINT "DIER NIET GERADEN, SPEL IS
AFGELOPEN"

```

Opdracht 23

```

10 REM **FAMILIE**
20 DIM A$(40),B$(40),C$(40)
30 DATA MOEDER,AN
40 DATA VADER,PETER
50 DATA ZUS,SUSAN
60 DATA BROER,MARC ERWIN ERIK
70 DATA OOM,JAN
80 DATA TANTE,RYAN MARIJKE
90 DATA NICHT,YVONNE
100 DATA NEEF,PASCAL
105 DATA NIETS,NIETS
110 PRINT "clear"
200 PRINT "WELK FAMILIELID WIL JE?"
210 INPUT C$
300 READ A$:READ B$
310 IF C$=A$ THEN PRINT "JE ";A$;"(S)
HEET/HETEN ";B$:PRINT :GOTO 320
312 IF A$="NIETS" THEN PRINT :PRINT "J
E HEBT GEEN ";C$:GOTO 320
315 GOTO 300
320 FOR T=1 TO 800:NEXT T
325 RESTORE :GOTO 30

```

Opdracht 24

```

10 REM >>ZO DENK IK OVER JOU<<
15 DIM A$(40),B$(40),C$(40),D$(40)
20 PRINT "clear"
25 PRINT "MIJN ONDERWIJZER(ES) HEET"

```

```
30 INPUT A$
32 PRINT
33 PRINT "ZO DENK IK OVER HEM/HAAR:"
34 PRINT
35 B$="IK VIND HEM/HAAR SOMS AARDIG"
36 C$="IK VIND HEM/HAAR EEN KLIER"
37 D$="HIJ/ZIJ IS DE BESTE!"
50 A=INT(RND(0)*3)+1
60 IF A=1 THEN PRINT B$
70 IF A=2 THEN PRINT C$
80 IF A=3 THEN PRINT D$
```

- ERROR- 2 Je hebt niet genoeg geheugen in je computer.
- ERROR- 3 De waarde die je hebt opgegeven is niet juist.
- ERROR- 4 Je hebt te veel variabelen gemaakt.
- ERROR- 5 Je hebt een fout gemaakt in de lengte van de 'tekst'.
- ERROR- 6 De beschikbare gegevens staan niet in het door jou gevraagde bereik. Deze foutmelding krijg je ook als je dingen in je programma hebt veranderd en je bent met de loper over het woord READY gekomen. Dit is gewoon, je hebt dus geen fout gemaakt.
- ERROR- 7 Het opgegeven getal is groter dan 32767.
- ERROR- 8 Foute INPUT instructie.
- ERROR- 9 Je hebt een DIM fout gemaakt of dezelfde DIM 2 keer gebruikt.
- ERROR- 10 Er staan te veel instructies in je programma. Kijk naar de GOTO en de FOR...NEXT instructies.
- ERROR- 12 De computer kan de aangewezen regel niet vinden.
- ERROR- 13 Je hebt geen passende FOR instructie gemaakt.
- ERROR- 14 De regel is te lang.
- ERROR- 15 FOR regel is verwijderd of bestaat niet.
- ERROR- 16 Je hebt een RETURN fout gemaakt.
- ERROR- 17 Je hebt een spellingsfout gemaakt in een instructie.
- ERROR- 18 Het eerste teken dat voor de 'tekst' staat is niet goed.

```
10 REM +++POPQUIZ+++
20 PRINT "clear"
30 DIM A$(30),B$(30),C$(30)
40 DATA ANNABEL,HANS DE BOOY
50 DATA DE BOM,DOE MAAR
60 DATA ZWART-WIT,FRANK BOEYEN GROEP
70 DATA HE AMSTERDAM,DRUKWERK
80 DATA BEAT IT,MICHAEL JACKSON
90 DATA YESTERDAY,THE BEATLES
100 DATA 99 LUFTBALLONS,NENA
110 DATA EEN BEETJE VAN DIT,VULCANO
120 DATA BREAKAWAY,TRACEY ULLMAN
130 DATA DOCTOR RYTHM,G'RACE
140 DATA HELLO,LIONEL RICHIE
150 DATA SHE'S A LIAR,THE DOLLY DOTS
160 DATA LADY,SPARGO
170 DATA VICTIMS,CULTURE CLUB
180 DATA RED RED WINE,UB40
300 PRINT "WIE ZINGTHET LIEDJE?":PRINT
310 R=INT(RND(0)*15)+1
320 FOR I=1 TO R
330 READ A$,B$:NEXT I
340 PRINT A$:PRINT:INPUT C$:PRINT
350 IF C$=B$ THEN PRINT"GOED":GOTO 365
360 PRINT "FOUT ";A$;" WORDT GEZONGEN
DOOR ";B$
365 FOR T=1 TO 1000:NEXT T
370 RESTORE:A$="":B$="":C$="":PRINT
"clear":GOTO 300
```

Het vorige programma heeft maar 15 DATA regels. Natuurlijk kun je deze naar hartelust uitbreiden of veranderen. Denk er dan wel aan dat je het getal 15 in regel 310 aanpast aan je eigen programma.

```

10 REM***WINTERSE SNEEUWBUI***
15 SETCOLOR 2,0,0
20 PRINT "clear"
30 POKE 752,1
40 FOR Y=0 TO 22
45 FOR T=0 TO 10
50 X=INT(RND(0)*40)
60 POSITION X,Y:PRINT "*"
70 NEXT T:NEXT Y
80 GOTO 40

```

```

10 REM===KLEURENSNOEPER===
20 GRAPHICS 3+16
30 SETCOLOR 1,1,8:SETCOLOR 2,11,10
40 SETCOLOR 3,6,8:SETCOLOR 4,9,10
50 FOR I=1 TO 38
60 FOR J=0 TO 22
70 COLOR INT(RND(0)*3)+1
80 PLOT I,J
90 NEXT J:NEXT I
100 FOR I=1 TO 38
110 FOR J=0 TO 22
120 REM EERSTE KLEUR WORDT OPGESNOEPT
130 LOCATE I,J,A:IF A=1 THEN COLOR 0
140 PLOT I,J
150 NEXT J:NEXT I
160 FOR I=1 TO 38
170 FOR J=0 TO 22
180 REM TWEDE KLEUR WORDT OPGESNOEPT
190 LOCATE I,J,A:IF A=2 THEN COLOR 0

```

```

200 PLOT J: NEXT I
210 NEXT J: NEXT I
220 FOR I=1 TO 38
230 FOR J=0 TO 22
240 REM NU NOG DE LAATSTE KLEUR
250 NEXT J: NEXT I
260 END
  
```

```

10 REM***GALGJE***
30 DIM A$(18), B$(1), C$(18)
35 GRAPHICS 0
40 PRINT "clear": P=0: A$=" ": C$="123456789
012345678"
50 PRINT "GALGJE IS EEN SPELLETJE"
60 PRINT :PRINT "VOOR TWEE PERSONEN"
70 PRINT :PRINT "SPELER TWEE KIJKT EVE
N NIET"
80 PRINT :PRINT "SPELER EEN TYPT EEN W
OORD IN"
90 PRINT :PRINT "VAN MAX. 18 LETTERS"
100 PRINT :INPUT A$
105 GRAPHICS 1:SETCOLOR 0,2,10
110 PRINT "clear"
120 PRINT "PROBEER HET WOORD LETTER VO
OR LETTER TE RADEN"
125 PRINT "TOETS 1 LETTER IN"
140 INPUT B$:P=P+1:R=0
150 FOR I=1 TO LEN(A$)
160 POSITION 1+I,9:IF B$=A$(I,I) THEN
PRINT #6;B$:C$(I,I)=B$:R=1
170 IF C$(I,I)<>A$(I,I) THEN Q=1
180 NEXT I
190 P=P-R
220 IF P=8 THEN PRINT "JE HEBT 8 KEER
MIS GERADEN, JE HANGT":FOR T=1 TO 650:
  
```

```

NEXT T:GOTO 250
230 IF Q=1 THEN PRINT :PRINT "PROBEER
NOG EEN LETTER":PRINT :Q=0:GOTO 140
240 GOTO 600
250 REM TEKENEN VAN DE GALG
260 GRAPHICS 8+16
270 SETCOLOR 2,12,8:SETCOLOR 1,12,0:CO
LOR 1
280 PLOT 120,185:DRAWTO 160,160:REM L.
POOT VAN DE GALG
300 DRAWTO 200,185:REM R.POOT GALG
310 PLOT 160,160:DRAWTO 160,30:REM VER
TICALE PAAL GALG
320 DRAWTO 220,30:REM HOR.LAT GALG
330 DRAWTO 220,60:REM TOUW
340 FOR A=0 TO 6.28 STEP 0.1:REM HOOFD
MANNETJE
350 X=220-10*SIN(A):Y=70-10*COS(A)
360 PLOT X,Y:DRAWTO X,Y
370 NEXT A
380 PLOT 220,80:DRAWTO 220,120:REM LIJF
MANNETJE

390 DRAWTO 210,160:REM L.BEEN
400 PLOT 220,120:DRAWTO 230,160:REM R.
BEEN
410 PLOT 205,100:DRAWTO 220,88:REM L.A
RM
420 DRAWTO 235,100:REM R.ARM
430 PLOT 215,67:PLOT 225,67:PLOT
220,70:REMOGENENNEUS
440 PLOT 217,75:DRAWTO 223,75:REM MOND
450 FOR T=1 TO 650:NEXT T
500 PRINT "DRUK OP START ALS JE NOG EEN
SPELLETJE WILT SPELEN":PRINT

```

```
510 IF PEEK(513279) <> 6 THEN GOTO 510:REM DE PEEK IS DE GEHEUGENDOOS VOOR DE S  
TART-TOETS  
520 GOTO 35  
600 POSITION 2,12:PRINT #6;"GERADEN":G  
OTO 500
```

```
10 REM *** RAAD MIJN LEEFTIJD ***  
20 PRINT "clear"  
30 POKE 710,0:POKE 712,0  
40 POKE 752,1:REM LOPER WEG  
100 PRINT :PRINT "DENK AAN JE LEEFTIJD  
"  
110 POSITION 2,22  
120 PRINT "DRUK OP RETURN ALS JE KLAAR  
BENT"  
130 POSITION 2,2  
135 GOSUB 500  
140 PRINT "VERDUBBEL HET"  
150 GOSUB 500  
160 A=INT(RND(0)*20)+1  
170 B=A+A  
180 PRINT "TEL ER ";B;" BIJ OP"  
190 GOSUB 500  
220 POKE 752,0:REM LOPER EVEN AAN  
225 PRINT "WELK GETAL HEB JE NU?"  
226 PRINT  
230 INPUT C:IF C<B THEN 600  
240 POKE 752,1:REM LOPER WEER UIT  
250 IF INT(C/2)<>C/2 THEN GOTO 600  
260 GOSUB 530  
300 PRINT "TREK JE LEEFTIJD HIERVAN AF  
"  
310 GOSUB 500  
320 PRINT "DEEL HET DOOR TWEE"
```

```

330 GOSUB 500
340 PRINT :PRINT "JE BENT ";(C-B)/2;" JAAR OUD"
400 FOR T=1 TO 800:NEXT T:GOTO 10
500 POKE 764,255
510 IF PEEK(764)<>12 THEN 510
520 FOR T=1 TO 40:NEXT T:POKE 764,255
530 POSITION 2,2
540 PRINT ""
550 POSITION 2,2
590 RETURN
600 PRINT :PRINT "DAT KAN LEKKER NIET.
..!!!"
610 FOR T=1 TO 1500:NEXT T
620 GOTO 10

```

```

10 REM GEHEIMSCHRIFT
15 PRINT "clear"
20 DIM A$(100),K$(2),L$(2),S$(100)
30 PRINT "DIT IS CODERINGS-PROGRAMMA"
40 PRINT "GEEF ME EEN ZIN "
41 PRINT "DAN ZET IK HEM OM IN CODE"
45 PRINT :INPUT A$
50 L=LEN(A$)
60 A$(L+1)=" "
65 FOR I=1 TO L STEP 2
70 K$=A$(I,I+1)
80 L$=K$(2)
90 L$(2)=K$(1,1)
100 S$(I)=L$
110 NEXT I
120 PRINT
130 PRINT "HIER KOMT DE ZIN IN CODE"
140 PRINT
150 PRINT " " :S$

```



```
10 REM <<ALFABETLISCH>>
12 PRINT "clear"
15 DIM WOORD$(30),ALFABET$(30)
20 PRINT "IK ZET DE LETTERS VAN JOUW W
OORD"
25 PRINT "NETJES OP ALFABET"
30 PRINT "KOM MAAR OP MET JE WOORD"
35 PRINT :PRINT :INPUT WOORD$
40 L=LEN(WOORD$)
45 A=1
50 FOR X=65 TO 65+26
55 REM TEST OM TE ZIEN WELKE LETTERS
60 REM VAN HET ALFABET IN HET WOORD ST
AAN
65 FOR R=1 TO L
70 G=ASC(WOORD$(R,R))
75 IF G=X THEN ALFABET$(A)=CHR$(G):A=A
+1
80 NEXT R:NEXT X:PRINT :PRINT :PRINT
85 PRINT "HIER KOMT JE WOORD"
90 PRINT :PRINT ALFABET$
```





Wolters-Noordhoff